

Immediate Analysis of Periodic Steady States in Switched DC-DC Converters via SPICE

Dalibor BIOLEK^{1,2}, Viera BIOLKOVÁ³, Zdeněk KOLKA³

¹ Dept. of Electrical Engineering, University of Defense Brno, Kounicova 65, 662 10 Brno, Czech Republic

² Dept. of Microelectronics, Brno University of Technology, Technická 10, 616 00 Brno, Czech Republic

³ Dept. of Radio Electronics, Brno University of Technology, Purkyňova 118, 612 00 Brno, Czech Republic

dalibor.biolek@unob.cz, biolkova@feec.vutbr.cz, kolka@feec.vutbr.cz

Abstract. *The method of immediate analysis of periodic steady states in switched DC-DC converters operating in the continuous current mode is described. The initial conditions, which correspond to the periodic steady state, are found in the first step. They are used consequently for the conventional transient analysis. A special SPICE model of the converter finds automatically these initial conditions, which are then available within the transient analysis. The method works both for the well-known behavioral models of switched converters and also for models which employ complex nonlinear SPICE models of semiconductor switches.*

Keywords

Switched DC-DC converter, averaged model, switching effects, periodic steady state.

1. Introduction

The analysis of periodic steady states (PSS) in switched DC-DC converters via SPICE-family programs belongs to rather problematic tasks [1]-[3]. Most of these programs do not provide the algorithms of automated steady-state computation. The transient analysis of waveforms, consisting of a mix of slow and fast phenomena, is time-consuming and it can also lead to accumulation of numerical errors. The initial transients in switched converters can persist over hundreds or even thousands of switching periods. This fact complicates, for example, the steady-state analysis in the well-known OrCAD PSpice programs. Note that the PSS waveforms are important from the viewpoint of reading key values such as the DC components of voltages and currents and powers and efficiency, the voltage- and current- ripple, and other parameters derived from these components.

Some special programs, for example HSpice RF [4] or Micro-Cap 10 [5], are equipped with algorithms fixing the periodic steady states. However, such algorithms are intended primarily for analyzing time-invariant systems, thus they do not work properly for switched circuits.

Averaging modeling of switched converters can speed-up their transient analysis considerably [6]-[8], and it can be also used for DC and AC analyses. These techniques enable, among other things, fast analysis of DC components of the PSS. However, the averaging models leave out fast phenomena coming from the switching processes. That is why we cannot get detailed waveforms which would give information about voltage ripple and other important parameters.

Other methods such as the method of generalized transfer functions [9]-[10] can model the switching effects but only in the frequency and not the time domain. Moreover, their implementation in the SPICE-family programs is not trouble-free [10].

For the above reasons, the steady-state analysis of switched DC-DC converters in SPICE is commonly performed via the successive and time-consuming transient analysis.

A method for the analysis of switching effects in DC-DC converters via DC bias point computation in SPICE is described in [11]. The reactance elements of the converter are replaced by their so-called DC equivalents. The coordinates of the PSS are obtained at four boundary points of the switching diagram through DC bias point computation. In [11], this method is described for simple behavioral models of switches where the active and passive switches are characterized by linear resistors, modeling the on-states. The open switches are modeled by open circuits.

The method of immediate PSS analysis, described here, consists of two phases. With respect to the fundamental limitations of SPICE, they must overlap each other. A special sub-model of switched converter, working on the basis of the method from [11] but generalized for nonlinear models of semiconductor switches (transistors and diodes), serves for predicting the PSS coordinates. The initial conditions of the PSS are extracted from this model in the form of state variables of the converter, i.e. capacitor voltage and inductor current. These initial conditions are implemented in the conventional SPICE model of the DC-DC converter via additional controlled sources. The following transient analysis is performed under the condition of si-

multaneous analysis of both models – the converter model and its special sub-model.

The paper is organized as follows: Section 2, which follows this Introduction, summarizes the main idea of the method of immediate computation of the PSS coordinates from [11], and its generalization for nonlinear switch models is given. The connection of the corresponding sub-model with the respective converter model is explained in Section 3. Section 4 demonstrates some results of the simulation of boost converter.

2. Immediate Computation of PSS Coordinates

Consider a DC-DC converter in the continuous current mode with a switching period T_s which is subdivided into the switching phases 1 and 2 of lengths dT_s and $d'T_s = (1-d)T_s$, where d is the duty ratio. Phase 1/2 corresponds to the active switch (transistor) in on/off state.

The method of immediate computation of the PSS coordinates, described in [11], is based on the assumption that the waveforms of the converter state variable derivatives with respect to time can be truly approximated by piece-wise-linear (PWL) functions according to Fig. 1. The state variables are here the capacitor voltages v_C and inductor currents i_L , and thus their derivatives are proportional to capacitor currents i_C and inductor voltages v_L . Such assumption means that the state variable waveforms can be truly approximated by parabolic curves. This hypothesis is based on the fact that a well-designed converter exhibits pretty long time constants of transients as a necessity of high-quality filtration of the voltage ripple. The good linearity of the above derivatives can be observed also in the case of inconveniently small and in practice untypical time constants [11]. A steady-state analysis based on such assumptions then leads to results with an accuracy which is acceptable for a large scale of the parameters of converter components.




Fig. 1. Piece-wise-linear approximation of the state variable derivatives with respect to time [11].

Four points, denoted 1-, 1+, 2-, and 2+, which correspond to left- and right-side limits at time instants of the end of switching phases 1 and 2, are characteristic of the steady-state waveforms in Fig. 1. Let us denote I_{C1+} , I_{C1-} , etc. the corresponding limits of the circuit variables. Then the following equations can be derived for the voltage of

capacitor with the capacitance C and for the current through inductor with the inductance L , considering the assumption of the PWL approximation [11]:

$$V_{C1-} = V_{C2+} + \frac{dT_s}{2C}(I_{C2+} + I_{C1-}), \quad (1)$$

$$I_{L1-} = I_{L2+} + \frac{dT_s}{2L}(V_{L2+} + V_{L1-}), \quad (2)$$

$$V_{C2-} = V_{C1+} + \frac{d'T_s}{2C}(I_{C1+} + I_{C2-}), \quad (3)$$

$$I_{L2-} = I_{L1+} + \frac{d'T_s}{2L}(V_{L1+} + V_{L2-}). \quad (4)$$

The state variables are continuous functions of time, thus $V_{C1-} = V_{C1+} = V_{C1}$, $V_{C2-} = V_{C2+} = V_{C2}$, $I_{L1-} = I_{L1+} = I_{L1}$, $I_{L2-} = I_{L2+} = I_{L2}$.

Equations (1) and (2) were derived via integration of the waveforms in Fig. 1 within switching phase 1. That is why they represent mathematical models of the capacitor and the inductor for this switching phase.




Fig. 2. (a) Symbolic representation of switched DC-DC converter which contains an active (transistor) and a passive (diode) switch, inductor, and capacitor; (b) sub-model for computing the PSS co-ordinates.

The corresponding model of the converter is demonstrated in Fig. 2 (b), upper section, with on-state active switch, open passive switch, the capacitor modeled via a controlled voltage source according to (1), and the in-

ductor modeled via a controlled current source according to (2). The DC bias point computation will give the circuit variables as limit values at point 1- (i.e. the left-side limits).

Accordingly, formulae (3) and (4) represent models of the capacitor and inductor in switching phase 2. By analyzing the corresponding converter model, the limit values of the circuit variables at point 2- can be obtained.

However, the right-side limits of the circuit variables also appear in the formulae for modeling the controlled sources. They can be obtained, with the help of the continuity of state variables, via solving another two sub-models of the converter. The sub-model in phase 1, where the capacitor and inductor are modeled as the controlled source of voltage V_{C2} , and current I_{L2} , provides the solution at point 2+. Similarly, the sub-model in phase 2 with the voltage source V_{C1} , instead of capacitor and with the current I_{L1} , replacing the inductor provides the solution at point 1+.

Subsequently, the four partial sub-models in Fig. 2 (b) are analyzed instead of the original switched-level model of the converter. Then a simple DC bias point calculation provides automatically the left- and right-side limits of each circuit variable at time instants which correspond to the modifications of the switch states. These limits are available as computed DC voltages and currents in the individual sub-model sections in Fig. 2 (b).

Note that it is not necessary to model the switches, used in the sub-models in Fig. 2 (b), by the simple behavioral method as described in [11]. The on-state active switch from sections no. 1 and 4 can be modeled directly by the nonlinear SPICE model of the transistor, with DC voltage of V_{on} level applied to its gate. The open active switch can be modeled in a similar way but with V_{off} level of the DC controlled voltage, or, more preferably, via removing the non-conducting MOSFET from the model. The latter method can be successfully used also for modeling the off-state diode. Omitting the corresponding models results in decreasing the computational cost and also in preventing potential convergence problems.

When limiting the analysis aims to the coordinates of the PPS points 1-, 1+, 2-, and 2+, and omitting the steady-state waveforms of the circuit variables as functions of continuous time, then it would be enough to compile only the sub-model in Fig. 2 (b), and to run the DC bias point computation. Note that SPICE computes the limit values of all circuit variables of the converter, not only those mentioned in (1) to (4). This is to say that, for example, the limit values of the output voltage or inductor current for computing the voltage and current ripples will also be available within a fraction of a second. Results obtained in this way can be affected by two types of error. The first one can be caused by violating the assumption of PWL waveforms in Fig. 1. The second source of errors can be in dynamic phenomena when switches change their states, which cannot be considered in the “DC models” (for example, logic feedthrough). Nevertheless, experience


shows that these errors are not significant and that the above procedure approaches the proximity of the real steady state. When the computed limit values are used as the initial conditions of the conventional transient analysis, then the possible deviations from the real steady state are corrected within several switching periods.

The next section gives an explanation of the mechanism of connecting the sub-model for computing the PSS coordinates from Fig. 1 (b) to the switch-level model of the converter.

3. Complete Model of Switched DC-DC Converter for Immediate PSS Analysis

When the initial conditions, for example capacitor voltages and inductor currents which correspond to the PSS, are known, they can be assigned to the IC attributes of capacitor and inductor SPICE models or entered as general initial conditions using the .IC command. Then the subsequent transient analysis would lead directly to the periodic steady state.

However, this procedure cannot be used directly because the initial conditions are present in the form of voltages and currents of the sub-model, computed via an iteration process, and they cannot be transferred to IC attributes. Such a transfer can be done indirectly via controlled sources, particularly via a current source in parallel to the inductor and via a voltage source in series with the capacitor. Such a co-operation of the auxiliary sub-model and the switched model of the converter will be explained on the example of the DC-DC buck converter in Fig. 3 [12]. The values of the parameters of passive elements were taken from [13].



$$V_g = 60V \quad R_{out} = 60 \text{ Ohms} \quad L = 6mH \quad C = 1000 \mu F$$

$$R_L = 3 \text{ Ohms} \quad R_c = 1 \text{ Ohm} \quad d = 0.25 \quad T_s = 100 \mu sec$$

Fig. 3. Switched boost converter, switch-level model with semiconductor switches [12].

The IRF045 transistor is controlled by rectangular waveform with levels 0 V and 8 V, 0.25 duty ratio, and a frequency of 10 kHz. The transient analysis of this converter was performed via switched-level and averaging

models in [12]. The results given in Fig. 4, adopted from [12], demonstrate that the converter approaches the steady state after more than 20 ms, which represents more than 200 switching periods. However, in order to obtain the steady-state waveforms with a reasonable precision, the simulation time should be extended to ca 500 periods. It is obvious that such a procedure is time-consuming and impracticable, particularly for a repetitive PSS analysis with various parameters of the circuit components.

The complete converter model in Fig. 5 contains the respective switched-level model, and the auxiliary four-section sub-model for computing the coordinates of PSS, which are transferred to the switched model in the form of initial conditions. As is obvious from the waveform of the switching signal, it is necessary to transfer the inductor current and capacitor voltage at time point 2+, which corresponds to the starting point of the transient analysis. Both models are solved simultaneously during this analysis, whereas the behavior of the auxiliary circuit is the same at each step. That is why the extracted values $I(G4)$ and $V(E4)$ behave like fixed initial conditions during the entire analysis run, and the analysis of the PSS is performed around them.




Fig. 4. Starting the converter from Fig. 3 to the PSS. Results of conventional transient analysis for switched (with emphasized switching effects) and averaged (with smoothed switching effects) converter model [12].




Fig. 5. Complete model of the converter for PSS transient analysis

4. Example of Simulation

The complete circuit file for PSPICE analysis, written on the basis of Fig. 5, is listed below:

```

*Boost converter - immediate steady state
.param Vg 60 RL 3 C 1000u Rc 1 Rout 60 L 6m
+ d 0.25 fs 10k
*immediate SS computation
*****
Vg1- in1- 0 {Vg}
RL1- in1- r11- {RL}
G1 r11- sw1- value={-I(Vg2+)+(V(r12+,sw2+)+
+ V(r11-,sw1-))*d/2/L/fs}
E1 out1- cr1- value={V(out2+,cr2+)+(I(E4)+I(E1))*
+ d/2/C/fs}
Rc1- cr1- 0 {Rc}
Rout1- out1- 0 {Rout}
Mswitch1- sw1- switch1- 0 0 IRF045
Vswitch1- switch1- 0 8V
*****
Vg1+ in1+ 0 {Vg}
RL1+ in1+ r11+ {RL}
G2 r11+ sw1+ value={-I(Vg1-)}
E2 out1+ cr1+ value={V(out1-,cr1-)}
Rc1+ cr1+ 0 {Rc}
Rout1+ out1+ 0 {Rout}
Diode1+ sw1+ out1+ MBRS140LT3_ON
*****
Vg2- in2- 0 {Vg}
RL2- in2- r12- {RL}
G3 r12- sw2- value={-I(Vg1+)+(V(r11+,sw1+)+
+ V(r12-,sw2-))*(1-d)/2/L/fs}
E3 out2- cr2- value={V(out1+,cr1+)+(I(E3)+I(E2))*
+ (1-d)/2/C/fs}
Rc2- cr2- 0 {Rc}
Rout2- out2- 0 {Rout}
Diode2- sw2- out2- MBRS140LT3_ON
*****
Vg2+ in2+ 0 {Vg}
RL2+ in2+ r12+ {RL}
G4 r12+ sw2+ value={-I(Vg2-)}
E4 out2+ cr2+ value={V(out2-,cr2-)}
Rc2+ cr2+ 0 {Rc}
Rout2+ out2+ 0 {Rout}
Mswitch2+ sw2+ switch2+ 0 0 IRF045
Vswitch2+ switch2+ 0 8V
*****
*end of immediate SS computation
*switched model with SS initial conditions
Vg in 0 {Vg}
RL in r1 {RL}
L r1 sw {L}
Gini r1 sw value={-I(Vg2-)}
C outS crx {C}
Eini crx cr value={V(out2+,cr2+)}
Rc cr 0 {Rc}
Rout outS 0 {Rout}
Mswitch sw switch 0 0 IRF045
    
```

```

Diode sw outS MBRS140LT3_ON
Vswitch switch 0 PULSE 0 8 0 10n 10n {d/fs} {1/fs}
*end of switched model with SS initial conditions
.lib
.tran 0 1m 0 10n skipbp
.probe
.end
    
```

The simulation was performed in OrCAD PSpice v. 16.1, with the results given in Fig. 6. The low value of the Step Ceiling (10ns) was chosen with the aim of having a detailed analysis of the spikes present in the output voltage, caused by the logic feedthrough. The follow-up analysis has shown that the algorithm identified the PSS perfectly.




Fig. 6. Results of the transient analysis via immediate transition to PSS: output voltage (top), inductor current (bottom).

5. Conclusions

The proposed method of modeling switched DC-DC converters speeds up considerably their steady-state transient analyses. When the condition of piece-wise-linear character of the waveforms of inductor voltage and capacitor current according to Fig. 1 is fulfilled, then the periodic steady state can be found not only "immediately" but also accurately. This method is useful for the analysis of switched converters of arbitrary topologies, working in the continuous current mode. The switches can be modeled with increased accuracy by nonlinear SPICE models of transistors and diodes rather than by commonly used simple and inaccurate behavioral models. A certain drawback is the necessity of building an auxiliary sub-model for computing the PSS coordinates. However, this discomfort is fully compensated by highly efficient transient analysis which is supported by the auxiliary four-section sub-model.

Acknowledgements

This work has been supported by the Czech Science Foundation under grant No. P102/10/1665, by the developing program of K217 UD Brno, and by the project CZ.1.07.2.3.00.20.0007 WICOMT, financed from the operational program Education for Competitiveness. The research leading to these results has received funding from the European Community's Seventh Framework Programme (FP7/2007-2013) under grant agreement No 230126. The described research was performed in laboratories supported by the SIX project; the registration number CZ.1.05/2.1.00/03.0072, the operational program Research and Development for Innovation.

References

- [1] VLACH, J., OPAL, A. Analysis and sensitivity of periodically switched linear networks. *IEEE Transactions on Circuits and Systems*, 1989, vol. 36, no. 4, p. 522-532.
- [2] YUAN, F., OPAL, A. *Computer Methods for Analysis of Mixed-Mode Switching Circuits*. Kluwer Academic Publishers, 2004.
- [3] BIOLEK, D., DOBEŠ, J. Computer simulation of continuous-time and switched circuits: Limitations of SPICE-family programs and pending issues. In *Proc. of Radioelektronika 2007*. Brno (Czech Republic), 2007, p. 15 – 25.
- [4] www.synopsys.com
- [5] www.spectrum-soft.com
- [6] MIDDLEBROOK, R. D., ČUK, S. A general unified approach to modeling switching-converter power stages. *Int. Journal of Electronics*, 1977, vol. 42, no. 6, p. 512-550.
- [7] ERICKSON, R. W. *Fundamentals of Power Electronics*. Cluwer Academic Publishers, 2004.
- [8] VORPÉRIAN, V. *Fast Analytical Techniques for Electrical and Electronic Circuits*. Cambridge University Press, 2002.
- [9] BIOLEK, D. Modeling of periodically switched networks by mixed s-z description. *IEEE Transactions on CAS-I*, 1997, vol. 44, no. 8, p. 750-758.
- [10] BIOLEK, D., BIOLKOVÁ, V., DOBEŠ, J. Modeling of switched DC-DC converters by mixed s-z description. In *Proc. of the IEEE International Symposium on Circuits and Systems, ISCAS2006*. Kos (Greece), 2006, p. 831 – 834.
- [11] BIOLEK, D., BIOLKOVÁ, V., KOLKA, Z. Analysis of switching effects in DC-DC converters via bias point computation. In *Proc. of the European Conference on Circuit Theory and Design, ECCTD07*. Sevilla (Spain), 2007, p. 1006-1009.
- [12] BIOLEK, D., BIOLKOVÁ, V., KOLKA, Z. Averaged modeling of switched DC-DC converters based on Spice models of semiconductor switches. In *Proc. of the 7th WSEAS Int. Conference on Circuits, Systems, Electronics, Control and Signal Processing, CSECS'08*. Tenerife (Spain), 2008, p. 162-167.
- [13] DIJK, E., SPRUIJT, J. N., SULLIVAN, M. O., KLAASSENS, J.B. PWM-switch modeling of DCDC converters. *IEEE Transactions on Power Electronics*, 1995, vol. 10, no. 6, 1995, p. 659-664.

About Authors...

Dalibor BIOLEK received the M.Sc. degree in Electrical Engineering from Brno University of Technology, Czech Republic, in 1983, and the Ph.D. degree in Electronics from the Military Academy Brno, Czech Republic, in 1989. He is currently with the Dept. of EE, University of Defense Brno (UDB), and with the Dept. of Microelectronics, Brno University of Technology (BUT), Czech Republic. His scientific activity is directed to the areas of general circuit theory, frequency filters, and computer simulation of electronic systems. For years, he has been engaged in algorithms of the symbolic and numerical computer analyses of electronic circuits with a view to the linear continuous-time and switched filters. He has published over 300 papers and is author of two books on circuit analysis and simulation. At present, he is professor at BUT and UDB in the field of Theoretical Electrical Engineering. Prof. Biolek is a member of the CAS/COM Czech National Group of IEEE.

Viera BIOLKOVÁ received her M.Sc. degree in Electrical Engineering from Brno University of Technology, Czech Republic, in 1983. She joined the Dept. of Radio Electronics in 1985. Her research and educational interests include signal theory, analog signal processing, and digital electronics.

Zdeněk KOLKA received the M.Sc. and Ph. D. degrees in 1992 and 1997 from Brno University of Technology, Czech Republic. Currently he is professor at the Dept. of Radio Electronics, Brno University of Technology. He reads lectures in the subjects "Computer-Aided Circuit Design" and "Computer and Communication Networks". Areas of his research interest include linear and nonlinear circuit modeling, numerical methods and circuit simulators. He was involved in the development of several methods for approximate symbolic analysis based on the simplification of circuit model.